<?xml version="1.0" encoding="UTF-8"?>

<!-- 
 * Copyright 2008 Philippe Prados.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 *
-->

<?xml-stylesheet type="text/xsl" href="../../../../xslt/xslt-to-xhtml.xslt" ?>
<xsl:stylesheet
	version="2.0"
	xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"
	xsi:schemaLocation=
		"http://www.w3.org/1999/XSL/Transform http://www.w3.org/2007/schema-for-xslt20.xsd
		 http://jndi-resources.googlecode.com/1.0/ http://www.prados.fr/xsd/1.0/jndi-resources.xsd
		"
	xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
	xmlns:jndi="http://jndi-resources.googlecode.com/1.0/"
	xmlns:tools="http://jndi-resources.googlecode.com/1.0/java/com.googlecode.jndiresources.tools.XSLTools"
	exclude-result-prefixes="#all"
>
<xsl:strip-space elements="*"/>
<xsl:output method="xml" indent="yes" encoding="utf-8" omit-xml-declaration="no"/>

<xsl:variable  name="basefilename"><xsl:value-of select="concat('deploy/',$currentid,'/',$currentid,'_activemq_cf-ds')"/></xsl:variable>
<xsl:variable  name="filename"><xsl:value-of select="concat($targetdir,'jboss.server.conf/',$basefilename,'.jndi')"/></xsl:variable>

<xsl:template match="jndi:resources[@id=$currentid]">
<xsl:value-of select="tools:info(concat('Generate ',$filename))"/>
<xsl:result-document  href="{$filename}">
<xsl:text>&#xA;</xsl:text>
<xsl:comment>Generated by jndi-resources v<xsl:value-of select="$version"/></xsl:comment>
<xsl:text>&#xA;</xsl:text>
<xsl:processing-instruction name="jndi-stylesheet">href="xslt/save.xslt" target="<xsl:value-of select="concat($basefilename,'.xml')"/>"</xsl:processing-instruction>
<!-- <xsl:processing-instruction name="jndi-stylesheet">href="xslt/merge-jboss-server.xslt" target="<xsl:value-of select="concat($basefilename,'_jms-service.xml')"/>"</xsl:processing-instruction>-->
<xsl:text>&#xA;</xsl:text>
<connection-factories>
	<xsl:apply-templates />
</connection-factories>
</xsl:result-document>
</xsl:template>

<xsl:template match="jndi:resources[@id=$currentid]/jndi:resource[@familly=$familly]">

	<xsl:if test="jndi:property[@name='xa' and @value='no']">
		<xsl:text>&#xA;   </xsl:text>
		<xsl:comment>Register JMS factory for key <xsl:value-of select="@name"/></xsl:comment>
		<xsl:text>&#xA;   </xsl:text>
		<no-tx-connection-factory>
			<use-java-context>true</use-java-context>
			<jndi-name><xsl:value-of select="concat('${jndi-prefix}',@name)"/></jndi-name>
			<track-connection-by-tx/>
			<rar-name>activemq-ra.rar</rar-name>
			<connection-definition>javax.jms.ConnectionFactory</connection-definition>
			<ServerUrl>${jms.activemq.url}</ServerUrl>
			<!--
			<UserName>sa</UserName>
			<Password></Password>
			-->
			<min-pool-size>1</min-pool-size>
			<max-pool-size>200</max-pool-size>
			<blocking-timeout-millis>30000</blocking-timeout-millis>
			<idle-timeout-minutes>3</idle-timeout-minutes>
			<xsl:apply-templates select="jndi:extends[@appsrv=$appsrv]/*" mode="remove-namespace"/>
		</no-tx-connection-factory>
	</xsl:if>
	<xsl:if test="jndi:property[@name='xa' and @value!='no'] or count(jndi:property[@name='xa'])=0">
		<xsl:text>&#xA;   </xsl:text>
		<xsl:comment>Register JMS factory for key <xsl:value-of select="@name"/></xsl:comment>
		<xsl:text>&#xA;   </xsl:text>
		<tx-connection-factory>
			<use-java-context>true</use-java-context>
			<jndi-name><xsl:value-of select="concat('${jndi-prefix}',@name)"/></jndi-name>
		    <xsl:choose>
				<xsl:when test="jndi:property[@name='xa' and @value='local']">
					<local-transaction/>
				</xsl:when>
				<xsl:otherwise>
					<xa-transaction/>
				</xsl:otherwise>
			</xsl:choose>
			<track-connection-by-tx/>
			<rar-name>activemq-ra.rar</rar-name>
			<connection-definition>javax.jms.ConnectionFactory</connection-definition>
			<ServerUrl>${jms.activemq.url}</ServerUrl>
			<!--
			<UserName>sa</UserName>
			<Password></Password>
			-->
			<min-pool-size>1</min-pool-size>
			<max-pool-size>200</max-pool-size>
			<blocking-timeout-millis>30000</blocking-timeout-millis>
			<idle-timeout-minutes>3</idle-timeout-minutes>

			<xsl:apply-templates select="jndi:extends[@appsrv=$appsrv]/*" mode="remove-namespace"/>
		</tx-connection-factory>
	</xsl:if>
</xsl:template>

<xsl:template match="text()|comment()" />

<!-- Remove name-space -->
<xsl:template match="/|comment()|processing-instruction()" mode="remove-namespace">
    <xsl:copy>
      <xsl:apply-templates mode="remove-namespace"/>
    </xsl:copy>
</xsl:template>
<xsl:template match="*" mode="remove-namespace">
    <xsl:element name="{local-name()}">
      <xsl:apply-templates select="@*|node()" mode="remove-namespace"/>
    </xsl:element>
</xsl:template>
<xsl:template match="@*" mode="remove-namespace">
    <xsl:attribute name="{local-name()}">
      <xsl:value-of select="."/>
    </xsl:attribute>
</xsl:template>

</xsl:stylesheet>
